Equations jusqu'au quatrième degré dans les nombres complexes

Alain Colmerauer

24 mai 2016

Nous sommes ici dans le domaine du corps des nombres complexes. Les polynômes, les équations sont tous sur les complexes. Toute lettre dont la nature n'est pas précisé est un nombre complexe.

1 Précisions sur les racines des polynomes et des équations

Un *polynôme (à une inconnue) de degré n* est une expression de la forme

$$a_n Z^n + a_{n-1} Z^{n-1} + \cdots + a_2 Z^2 + a_1 Z + a_0$$

avec $a_n \neq 0$ et Z une variable. Les *racines de ce polynôme* sont les nombres complexes $z_1, z_2, \ldots, z_n$ tels que, quelque soit le nombre complexe z,

$$a_0 z^n + a_1 z^{n-1} + \cdots + a_n z^0 = a_0(z - z_1)(z - z_2) \cdots (z - z_n).$$

Le nombre de fois que la racine z_i apparaît à droite de l'égalité est sa *multiplicité*. Ces racines existent toujours dans le corps des nombres complexes. Ceci a été montré par Carl Gauss. On a

$$\begin{aligned}
z_1 + z_2 + \cdots + z_n &= (-1)^1 \frac{a_{n-1}}{a_n}, \\
z_1 z_2 + z_1 z_2 + \cdots + z_{n-1} z_n &= (-1)^2 \frac{a_{n-2}}{a_n}, \\
z_1 z_2 z_3 + z_1 z_2 z_4 + \cdots + z_{n-2} z_{n-1} z_n &= (-1)^3 \frac{a_{n-2}}{a_n}, \\
&\vdots \\
z_1 z_2 \cdots z_n &= (-1)^n \frac{a_0}{a_n}
\end{aligned}$$

Une expression de la forme

$$P(Z) = 0,$$

où $P(Z)$ est un polynôme de degré n sur la variable Z, est une *équation de degré n*. Une *solution de l'équation* est un a tel que $P(a) = 0$. On montre l'équivalence :

$$P(a) = 0 \quad \Leftrightarrow \quad a \text{ est une racine du polynome } P(Z)$$

Nous discourons *des racines de l'équation* pour les racines du polynôme associé.

2 Racines n-ième des nombres complexes

Donnons quelques précisions sur les nombres complexes. Grâce à la formule d'Euler et de Moivre,

$$e^{i\varphi} = \cos\varphi + i\sin\varphi,$$

tout nombre complexe non nul peut s'écrire de deux façons

$$x + iy \quad = \quad \rho\, e^{i}\varphi$$

avec x, y, ρ, φ des nombres réels et $0 \leq \varphi < 2\pi$. Les formules de conversion sont

$$x = \rho\cos\varphi,\ y = \rho\sin\varphi \quad \text{et} \quad \rho = \sqrt{x^2 + y^2},\ \{\varphi\} = [0, 2\pi[\, \cap \cos^{-1}\frac{x}{\rho} \cap \sin^{-1}\frac{y}{\rho}$$

Soit a un nombre complexe. Une *n-ième racine de a* est une racine du polynome

$$Z^n - a$$

L'opération $\sqrt[n]{\ }$ est étendue des réels aux complexes comme suit :

$$\sqrt[n]{\rho e^{i\varphi}} = \sqrt[n]{\rho}\, e^{i\frac{\varphi}{n}}$$

Nous continurons à écrire $\sqrt{\ }$ pour $\sqrt[2]{\ }$. Le nombre complexe $\sqrt[n]{a}$ est une racine n-ième de a. Si z_0 est une racine n-ième de a alors $z_0, z_1, \ldots, z_{n-1}$ avec

$$z_k = z_0 u_k, \quad u_k = e^{i\frac{2k\pi}{n}} = \left(e^{i\frac{2\pi}{n}}\right)^k$$

sont toutes les racines n-ième de a. Les u_k sont les racines n-ième de l'unité.

Remarqe 1
La valeur de $e^{i\frac{2\pi}{n}}$ pour $n = 2, 3$ est

$$e^{i\frac{2\pi}{2}} = -1, \quad \text{et} \quad e^{i\frac{2\pi}{3}} = -\frac{1}{2} + i\frac{\sqrt[2]{3}}{2}$$

et les racines n-ième de l'unité pour $n = 2, 3$ sont

$$u_0 = 1,\ u_1 = -1 \quad \text{et} \quad u_0 = 1,\ u_1 = -\frac{1}{2} + i\frac{\sqrt{3}}{2},\ u_3 = -\frac{1}{2} - i\frac{\sqrt{3}}{2}$$

Remarque 2
Attention les égalités

$$\sqrt[n]{z^n} = \left(\sqrt[n]{z}\right)^n \quad \text{et} \quad \sqrt[n]{z_1 z_2} = \sqrt[n]{z_1}\,\sqrt[n]{z_2}$$

ne sont pas toujours vrais dans cette extension de $\sqrt[n]{\ }$. Par exemple

$$\sqrt{(-1)^2} = 1,\ \left(\sqrt{-1}\right)^2 = -1 \quad \text{et} \quad \sqrt{(-1)(-1)} = \sqrt{1} = 1,\ \sqrt{-1}\sqrt{-1} = -1$$

3 Equation du premier et deuxième degré

Soit à résoudre l'équation du 1-ier degré $aZ + b = 0$ où $a \neq 0$. Ceci est complément évident. La racine de cet équation est z_0 avec

$$z_0 = -\frac{b}{a}$$

Soit à résoudre l'équation du 2-ième degré

$$aZ^2 + bZ + c = 0$$

où $a \neq 0$. C'est bien connu mais nous le refaisons. Posons $Z = X + t$. Donc nous avons

$$aX^2 + (2at + b)X + (at^2 + bt + c) = 0$$

En prenant

$$t = -\frac{b}{2a}, \quad p = \frac{at^2 + bt + c}{a} = -\left(\tfrac{b}{2a}\right)^2 + \tfrac{c}{a}$$

il suffit de résoudre

$$X^2 + p = 0$$

ce qui donne x_0, x_1 avec

$$x_k = (-1)^k \sqrt{-p}$$

Donc les racines de $aZ^2 + bZ + c = 0$ sont z_0, z_1 avec

$$z_k = -\frac{b}{2a} + (-1)^k \sqrt{\left(\frac{b}{2a}\right)^2 - \frac{c}{a}}$$

4 Equation du troisième degré : méthode de Cardan

Soit à résoudre l'équation du 3-ième degré

$$aZ^3 + bZ^2 + cZ + d = 0$$

où $a \neq 0$. Posons $Z = X + t$ donc

$$aX^3 + (3at + b)X^2 + (3at^2 + 2bt + c)X + (at^3 + bt^2 + ct + d) = 0$$

En prenant

$$t = -\frac{b}{3a}, \quad p = \frac{3at^2 + 2bt + c}{a}, \quad q = \frac{at^3 + bt^2 + ct + d}{a}$$

il suffit de résoudre

$$X^3 + pX + q = 0$$

ce que nous allons faire.

Soit à résoudre

$$Z^3 + pZ + q = 0$$

En posant $Z = u + v$ nous avons

$$u^3 + 3u^2v + 3uv^2 + v^3 + p(u + v) + q = 0$$

c'est-à-dire

$$u^3 + 3uv(u+v) + v^3 + p(u+v) + q = 0$$

c'est-à-dire

$$u^3 + (3uv+p)(u+v) + v^3 + q = 0$$

En prenant

$$uv = -\frac{p}{3}$$

on a

$$u^3 + v^3 = -q$$

mais aussi

$$u^3v^3 = -\left(\frac{p}{3}\right)^3$$

Donc u^3 et v^3 sont racines de l'équation

$$X^2 + qX - \left(\frac{p}{3}\right)^3 = 0$$

ce qui donne

$$u^3 = -\frac{q}{2} + \sqrt{\left(\frac{q}{2}\right)^2 + \left(\frac{p}{3}\right)^3}$$

$$v^3 = -\frac{q}{2} - \sqrt{\left(\frac{q}{2}\right)^2 + \left(\frac{p}{3}\right)^3}$$

Première façon

D'où l'on déduit, compte tenue de l'égalité $uv = -\dfrac{p}{3}$:

Les racines de l'équation $Z^3 + pZ + q = 0$ sont z_0, z_1, z_2 avec $z_k = u_k + v_k$ et $l \in \{0,1,2\}$ et

$$u_k = j^k \sqrt[3]{-\frac{q}{2} + \sqrt{\left(\frac{q}{2}\right)^2 + \left(\frac{p}{3}\right)^3}}, \quad v_k = j^\ell \sqrt[3]{-\frac{q}{2} - \sqrt{\left(\frac{q}{2}\right)^2 + \left(\frac{p}{3}\right)^3}}, \quad u_k v_k = -\frac{p}{3},$$

en posant $j = -\frac{1}{2} + i\frac{\sqrt{3}}{2}$.

Deuxième façon

Compte tenue de l'égalité $uv = -\dfrac{p}{3}$, nous avons $v = -\dfrac{p}{3u}$ si $p \neq 0$. Donc :

Si $p \neq 0$, les racines de l'équation $Z^3 + pZ + q = 0$ sont z_0, z_1, z_2 avec

$$z_k = u_k - \frac{p}{3u_k}, \quad u_k = j^k \sqrt[3]{-\frac{q}{2} + \sqrt{\left(\frac{q}{2}\right)^2 + \left(\frac{p}{3}\right)^3}}$$

Si $p = 0$ les racines de l'équation $Z^3 + q = 0$ sont z_0, z_1, z_2 avec $z_k = j^k\sqrt[3]{-q}$.

5 Equation du quatrième degré : formule de Lagrange

Soit à résoudre l'équation du 4-ième degré

$$aZ^4 + bZ^3 + cZ^2 + dZ + e = 0$$

où $a \neq 0$. Posons $Z = X + t$ donc

$$aX^4 + (4at + b)X^3 + (6at^2 + 3bt + c)X^2 + (4at^3 + 3bt^2 + 2ct + d)X + (at^4 + bt^3 + ct^2 + dt + e) = 0$$

En prenant

$$t = -\frac{b}{4a}, \quad p = \frac{6at^2 + 3bt + c}{a}, \quad q = \frac{4at^3 + 3bt^2 + 2ct + d}{a}, \quad r = \frac{at^4 + bt^3 + ct^2 + dt + e}{a}$$

il suffit de résoudre

$$X^4 + pX^2 + qX + r = 0$$

ce que nous allons faire.

Soit à résoudre

$$Z^4 + pZ^2 + qZ + r = 0$$

Dans cette optique, on pose alors :

$$\begin{aligned} y_1 &= -(z_1 + z_2)(z_3 + z_4) \\ y_2 &= -(z_1 + z_3)(z_2 + z_4) \\ y_3 &= -(z_1 + z_4)(z_2 + z_3) \end{aligned}$$

On trouve après de longs calculs fastidieux que nous relatons à la section suivante :

$$\begin{aligned} y_1 + y_2 + y_3 &= -2(z_1z_2 + z_1z_3 + z_1z_4 + z_2z_3 + z_2z_3 + z_3z_4) \\ y_1y_2 + y_1y_3 + y_2y_3 &= (z_1z_4 + z_2z_3 + z_1z_3 + z_2z_4 + z_1z_2 + z_3z_4)^2 - 4z_1z_2z_3z_4 \\ y_1y_2y_3 &= (z_1z_2z_3 + z_1z_2z_4 + z_1z_3z_4 + z_2z_3z_4)^2 \end{aligned}$$

donc

$$\begin{aligned} y_1 + y_2 + y_3 &= -2p \\ y_1y_2 + y_1y_3 + y_2y_3 &= p^2 - 4r \\ y_1y_2y_3 &= q^2 \end{aligned}$$

Nous voyons alors que y_1, y_2, y_3 sont les trois racines de l'équation :

$$Y^3 + 2pY^2 + (p^2 - 4r)Y - q^2 = 0$$

Cette équation étant du troisième degré, nous avons appris à la résoudre et nous en déduisons les valeurs de y_1, y_2, y_3 que l'on supposera désormais connues. À partir de ces valeurs, il nous faut maintenant en déduire les valeurs de z_1, z_2, z_3, z_4. Du fait que

$$z_1 + z_2 + z_3 + z_4 = 0$$

nous avons

$$\begin{aligned} y_1 &= (z_1 + z_2)^2 = (z_3 + z_4)^2 \\ y_2 &= (z_1 + z_3)^2 = (z_2 + z_3)^2 \\ y_3 &= (z_1 + z_4)^2 = (z_2 + z_3)^2 \end{aligned}$$

et donc

$$\begin{aligned}
z_1 + z_2 &= +\alpha\sqrt{y_1} && (1)\\
z_3 + z_4 &= -\alpha\sqrt{y_1} && (2)\\
z_1 + z_3 &= +\beta\sqrt{y_2} && (3)\\
z_2 + z_4 &= -\beta\sqrt{y_2} && (4)\\
z_1 + z_4 &= +\gamma\sqrt{y_3} && (5)\\
z_2 + z_3 &= -\gamma\sqrt{y_3} && (6)
\end{aligned}$$

avec $\alpha, \beta, \gamma \in \{1, -1\}$.

En additionnant membre à membre les équations (1), (3) et (5), nous obtenons :

$$2z_1 + z_1 + z_2 + z_3 + z_4 = \alpha\sqrt{y_1} + \beta\sqrt{y_2} + \gamma\sqrt{y_3}$$

et comme $z_1 + z_2 + z_3 + z_4 = 0$ il nous reste :

$$z_1 = \frac{1}{2}(+\alpha\sqrt{y_1} + \beta\sqrt{y_2} + \gamma\sqrt{y_3})$$

En additionnant membre à membre les équations (1), (4) et (6), nous obtenons de la même manière :

$$z_2 = \frac{1}{2}(+\alpha\sqrt{y_1} - \beta\sqrt{y_2} - \gamma\sqrt{y_3})$$

En additionnant membre à membre les équations (2), (3) et (6), nous obtenons de la même manière :

$$z_3 = \frac{1}{2}(-\alpha\sqrt{y_1} + \beta\sqrt{y_2} - \gamma\sqrt{y_3})$$

En additionnant membre à membre les équations (2), (4) et (5), nous obtenons de la même manière :

$$z_3 = \frac{1}{2}(-\alpha\sqrt{y_1} - \beta\sqrt{y_2} + \gamma\sqrt{y_3})$$

Si on donne à α, β, γ toutes les valeurs possibles et qu'on permute les indices des z_k on constate qu'il n'y a que deux seuls jeux de 4 valeurs possibles pour les z_k. Les voici :

$$z_k = \varepsilon u_k, \quad \varepsilon \in \{1, -1\}$$

et

$$\begin{aligned}
u_1 &= \tfrac{1}{2}(+\sqrt{y_1} + \sqrt{y_2} + \sqrt{y_3})\\
u_2 &= \tfrac{1}{2}(+\sqrt{y_1} - \sqrt{y_2} - \sqrt{y_3})\\
u_3 &= \tfrac{1}{2}(-\sqrt{y_1} + \sqrt{y_2} - \sqrt{y_3})\\
u_4 &= \tfrac{1}{2}(-\sqrt{y_1} - \sqrt{y_2} + \sqrt{y_3})
\end{aligned}$$

Il reste à déterminer ε. Les conditions pour que z_1, z_2, z_3, z_4 soient les solutions de $Z^4 + pZ^2 + qZ + r = 0$ sont

$$\begin{aligned}
z_1 + z_2 + z_3 + z_4 &= 0 && (7)\\
z_1z_2 + z_1z_3 + z_1z_4 + z_2z_3 + z_2z_3 + z_3z_4 &= p && (8)\\
z_1z_2z_3 + z_1z_2z_4 + z_1z_3z_4 + z_2z_3z_4 &= -q && (9)\\
z_1z_2z_3z_4 &= r && (10)
\end{aligned}$$

Seul la condition (9) est utililisable ici. Elle devient

$$\varepsilon(u_1u_2u_3 + u_1u_2u_4 + u_1u_3u_4 + u_2u_3u_4) = -q$$

Donc après une permutation éventuelle des indices de z_k :

les racines de l'équation $Z^4 + pZ^2 + qZ + r = 0$ sont

$$\begin{aligned} z_1 &= \varepsilon u_1, \quad u_1 = \frac{1}{2}(+\sqrt{y_1} + \sqrt{y_2} + \sqrt{y_3}) \\ z_2 &= \varepsilon u_2, \quad u_2 = \frac{1}{2}(+\sqrt{y_1} - \sqrt{y_2} - \sqrt{y_3}) \\ z_3 &= \varepsilon u_3, \quad u_3 = \frac{1}{2}(-\sqrt{y_1} - \sqrt{y_2} + \sqrt{y_3}) \\ z_4 &= \varepsilon u_4, \quad u_4 = \frac{1}{2}(-\sqrt{y_1} + \sqrt{y_2} - \sqrt{y_3}) \end{aligned}$$

où $\varepsilon \in \{1, -1\}$ avec

$$\varepsilon = \begin{cases} 1 \text{ (ou bien } -1), & \text{si } q = 0, \\ -\frac{1}{q}(u_1u_2u_3 + u_1u_2u_4 + u_1u_3u_4 + u_2u_3u_4), & \text{si } q \neq 0, \end{cases}$$

et où y_1, y_2, y_3 sont les trois racines de l'équation

$$y^3 + 2py^2 + (p^2 - 4r)y - q^2 = 0$$

Calculs fastidieux

Pour clarifier les calculs nous remplaçons les variables indicées z_1, z_2, z_3, z_4 par les lettres a, b, c, d. On a, en tenant compte notamment de $a + b + c + d = 0$:

$$\begin{aligned} y_1 &= -(a+b)(c+d) = -(ac+ad+bc+bd) = (a+b)^2 = a^2+b^2+2ab \\ y_2 &= -(a+c)(b+d) = -(ab+ad+bc+cd) = (a+c)^2 = a^2+c^2+2ac \\ y_3 &= -(a+d)(b+c) = -(ab+ac+bd+cd) = (b+c)^2 = b^2+c^2+2bc \end{aligned}$$

Premièrement :

$$y_1 + y_2 + y_3 \quad = \quad -2(ab+ac+ad+bc+bd+cd)$$

Deuxièmement, en tenant compte de $a + b + c + d = 0$:

$$\begin{aligned} y_1y_2 + y_1y_3 + y_2y_3 \quad = \quad & (a^2+b^2+2ab)(a^2+c^2+2ac) + \\ & (a^2+b^2+2ab)(b^2+c^2+2bc) + \\ & (a^2+c^2+2ac)(b^2+c^2+2bc) \\ = \quad & a^4 + (a^2b^2+a^2c^2+b^2c^2) + (2a^3b+2a^3c) + (2abc^2+2ab^2c) + 4a^2bc + \\ & b^4 + (a^2b^2+a^2c^2+b^2c^2) + (2ab^3+2b^3c) + (2abc^2+2a^2bc) + 4ab^2c + \\ & c^4 + (a^2b^2+a^2c^2+b^2c^2) + (2ac^3+2bc^3) + (2ab^2c+2a^2bc) + 4abc^2 + \end{aligned}$$

$$
\begin{aligned}
&= a^4+b^4+c^4+\\
&\quad 3(a^2b^2+a^2c^2+b^2c^2)+\\
&\quad 2(a^3b+a^3c+ab^3+b^3c+ac^3+bc^3)+\\
&\quad 4(a^2bc+ab^2c+abc^2)+\\
&\quad 4(a^2bc+ab^2c+abc^2)+\\
&= (a^4+b^4+c^4)+(a^2b^2+a^2c^2+b^2c^2)+\\
&\quad 2(a^2b^2+a^2c^2+b^2c^2)+\\
&\quad 2(a^2bc+ab^2c+abc^2)+2(a^3b+a^3c+ab^3+b^3c+ac^3+bc^3)+\\
&\quad 2(a^2bc+ab^2c+abc^2)+\\
&\quad 4(a^2bc+ab^2c+abc^2)\\
&= (a^2+b^2+c^2+ab+ac+bc)^2+4(a^2bc+ab^2c+abc^2)\\
&= (ab+ac+bc+(a+b+c)(-a-b-c))^2-4abc(-a-b-c)\\
&= (ab+ac+bc+ad+bd+cd)^2-4abcd
\end{aligned}
$$

Troisièmement, toujours en tenant compte de $a+b+c+d=0$:

$$
\begin{aligned}
y_1y_2y_3 &= (a^2+b^2+2ab)(a^2+c^2+2ac)(b^2+c^2+2bc)\\
&= 8a^2b^2c^2\\
&\quad 4(a^3bc^2+ab^3c^2)+4(a^3b^2c+ab^2c^3)+4(a^2b^3c+a^2bc^3)+\\
&\quad 2(a^4bc+ab^4c+abc^4)+\\
&\quad 2(a^3b^3+a^3c^3+b^3c^3)+\\
&\quad 2(a^3b^2c+a^2b^3c+a^3bc^2+a^2bc^3+ab^3c^2+ab^2c^3)+\\
&\quad (a^4b^2+a^2b^4+a^4c^2+a^2c^4+b^4c^2+b^2c^4)+2a^2b^2c^2\\
&= 10a^2b^2c^2\\
&\quad 2(a^3b^3+a^3c^3+b^3c^3)+\\
&\quad 2(a^4bc+ab^4c+abc^4)+\\
&\quad a^4b^2+a^2b^4+a^4c^2+a^2c^4+b^4c^2+b^2c^4+\\
&\quad 6(a^3b^2c+a^2b^3c+a^3b^2c+a^2b^3c+ab^3c^2+ab^2c^3)\\
&= 2^2a^2b^2c^2+\\
&\quad a^4b^2+a^4c^2+a^2b^4+b^4c^2+a^2c^4+b^2c^4+\\
&\quad 2\times(a^3b^2c+a^2b^3c+a^3b^2c+a^2b^3c+ab^3c^2+ab^2c^3)+\\
&\quad 2\times(a^4bc+ab^4c+abc^4)+\\
&\quad 2\times 3a^2b^2c^2+\\
&\quad 2\times(a^3b^3+a^3c^3+b^3c^3)+\\
&\quad 2\times 2(a^3b^2c+a^2b^3c+a^3bc^2+a^2bc^3+ab^3c^2+ab^2c^3)\\
&= [2abc+(a^2b+a^2c)+(ab^2+b^2c)+(ac^2+bc^2)]^2\\
&= [abc-(a^2b+a^2c+abc)-(ab^2+abc+b^2c)-(abc+ac^2+bc^2)]^2\\
&= [abc+(ab+ac+bc)(-a-b-c)]^2\\
&= (abc+abd+acd+bcd)^2
\end{aligned}
$$

6 Exemples

Voici des exemples que nous avons résolus avec le programme Prolog donné à la suite.

Exemple 1

$z^4 - 10z^3 + 35z^2 - 50z + 24 = 0$
$z_1 = 1, z_2 = 2, z_3 = 3, z_4 = 4$

```
?- racine([[1,0],[-10,0],[35,0],[-50,0], [24,0]],Z).
Z = [4.000000007450581, 0.0] ;
Z = [2.9999999925494194, 0.0] ;
Z = [1.9999999925494194, 0.0] ;
Z = [1.0000000074505806, 0.0] ;
false.
```

Exemple 2

$z^4 - 10z^3 + 35z^2 - 50z + 24 = 0$
$z_1 = 3 + 2i, z_2 = 5 + i, z_3 = i, z_4 = 1 + i$

```
?- racine([[1,0],[-9,-5],[14,33],[24,-44],[-26, 0]],Z).
Z = [3.0000000000000027, 2.000000000000001] ;
Z = [5.0000000000000036, 0.9999999999999939] ;
Z = [-4.440892098500626e-15, 1.0000000000000024] ;
Z = [0.9999999999999976, 1.000000000000003] ;
false.
```

7 Programme Prolog

```
/* Liste des racines d'un polynome */

racines(P,L) :- findall(Z,racine(P,Z),L).

/* Racine d'un polynome */

racine([A,B],Z) :- est(Z, moins(div(B,A))).
racine([A,B,C],Z) :-
   est(P, div(B,fois([2,0],A))),
   est(Q, div(C,A)),
   est(Z, add(moins(P),fois(racarreeun,racine(2,moins(carre(P),Q))))).
racine([A,B,C,D],Zp) :-
   est(T, div(B,fois([-3,0],A))),
   est(P, div(add(fois([3,0],fois(A,carre(T))),add(fois([2,0],fois(B,T)),C)),A)),
   est(Q, div(add(fois(A,cube(T)),add(fois(B,carre(T)),add(fois(C,T),D))),A)),
   solutionCardan(P,Q,Z),
   est(Zp, add(Z,T)).
racine([A,B,C,D,E],Zp) :-
   est(T, div(B,fois([-4,0],A))),
   est(P, div(add(fois([6,0],fois(A,carre(T))),add(fois([3,0],fois(B,T)),C)),A)),
   est(Q, div(add(fois(fois([4,0],A),cube(T)),add(fois(fois([3,0],B),carre(T)),add(fois(fois
   est(R, div(add(fois(A,pquatre(T)),add(fois(B,cube(T)),add(fois(C,carre(T)),add(fois(D,T),E
```

```
   solutionLagrange(P,Q,R,Z),
   est(Zp, add(Z,T)).

/* Polynome a partir de ses racines */

polynome(L,P) :- polynome(L,[[1,0]],P).

polynome([],P0,P0).
polynome([X|L],P0,P4) :-
   conc(P0,[[0,0]],P1),
   est(Xp, moins(X)),
   foisl(Xp,P0,P2),
   addll(P1,[[0,0]|P2],P3),
   polynome(L,P3,P4).

conc([],L,L).
conc([E|L],Lp,[E|Lpp]) :- conc(L,Lp,Lpp).

foisl(X,[],[]).
foisl(X,[Y|L],[Z|Lp]) :- est(Z, fois(X,Y)), foisl(X,L,Lp).

addl(X,[],[]).
addl(X,[Y|L],[Z|Lp]) :- est(Z, addl(X,Y)), addl(X,L,Lp).

addll([],[],[]).
addll([X|L],[Y|Lp],[Z|Lpp]) :- est(Z, add(X,Y)), addll(L,Lp,Lpp).

/* Solution de l'equation du troisieme degre selon Cardan */

solutionCardan(P,Q,Z) :-
   nul(P),
   est(Z, fois(racubiqueun,racine(3,moins(Q)))).
solutionCardan(Pp,Qp,Z) :-
   nonnul(Pp),
   est(P, div(Pp,[3,0])),
   est(Q, div(Qp,[2,0])),
   est(Raccubique, fois(racubiqueun,racine(3,moins(racine(2,add(carre(Q),cube(P))),Q)))),
   est(Z, moins(Raccubique,div(P,Raccubique))).

/* Solutions de l'equation du quatrieme degre selon Lagrange */

solutionLagrange(P,Q,R,Z) :-
   est(A, [1,0]),
   est(B, fois([2,0],P)),
   est(C, moins(carre(P),fois([4,0],R))),
   est(D, moins(carre(Q))),
   racines([A,B,C,D],[Y1,Y2,Y3]),
   est(Y1p, racine(2,Y1)),
```

```
   est(Y2p, racine(2,Y2)),
   est(Y3p, racine(2,Y3)),
   est(U1, div(add(Y1p,add(Y2p,Y3p)),[2,0])),
   est(U2, div(moins(Y1p,add(Y2p,Y3p)),[2,0])),
   est(U3, div(moins(Y3p,add(Y1p,Y2p)),[2,0])),
   est(U4, div(moins(Y2p,add(Y1p,Y3p)),[2,0])),
   est(V1, fois(U1,fois(U2,U3))),
   est(V2, fois(U1,fois(U2,U4))),
   est(V3, fois(U1,fois(U3,U4))),
   est(V4, fois(U2,fois(U3,U4))),
   epsilon(E, moins(add(V1,add(V2,add(V3,V4)))), Q),
   dans(U,[U1,U2,U3,U4]),
   est(Z, fois(E,U)).

epsilon([1,0],S,Q) :- nul(Q).
epsilon(E,S,Q) :- nonnul(Q), est(E, div(S,Q)).

dans(U,[U|L]).
dans(U,[Up|L]) :- dans(U,L).

/* Valeurs de l'enchainement des operations sur les complexes */

est(Z,Z) :- Z = [Z1,Z2].
est(Z,T) :- T =.. [F], Tp =.. [F,Z], call(Tp).
est(Z,T) :-
   T =.. [F,X], est(Xp,X), Tp =.. [F,Xp,Z], call(Tp).
est(Z,T) :-
   T =.. [F,X,Y], dif(F,racine), dif(F,'.'),
   est(Xp,X), est(Yp,Y), Tp =.. [F,Xp,Yp,Z], call(Tp).
est(Z,racine(N,X)) :- est(Xp,X), racine(N,Xp,Z).

/* Operations sur les complexes */

moins([X1,X2],[Y1,Y2]) :-  Y1 is -X1, Y2 is -X2.
moins([X1,X2],[Y1,Y2],[Z1,Z2]) :-  Z1 is X1-Y1, Z2 is X2-Y2.
add([X1,X2],[Y1,Y2],[Z1,Z2]) :-  Z1 is X1+Y1, Z2 is X2+Y2.
fois([X1,X2],[Y1,Y2],[Z1,Z2]) :- Z1 is X1*Y1-X2*Y2, Z2 is X1*Y2+X2*Y1.
inverse([X1,X2],[Y1,Y2]) :-
   Y1 is X1/(X1**2+X2**2), Y2 is -X2/(X1**2+X2**2).
div(X,Y,Z) :- inverse(Y,Yp), fois(X,Yp,Z).
carre(X,Y) :- fois(X,X,Y).
cube(X,Y) :- carre(X,Xp), fois(X,Xp,Y).
pquatre(X,Y) :- carre(X,Xp), carre(Xp,Y).

racarreeun([1,0]).
racarreeun([-1,0]).

racubiqueun([1,0]).
```

```
racubiqueun([X,Y]) :- X is -1/2, Y is sqrt(3)/2.
racubiqueun([X,Y]) :- X is -1/2, Y is -sqrt(3)/2.

racine(N,X,[0,0]) :- nul(X).
racine(N,X,Y) :-
   nonnul(X),
   polaire(X,[R,T]),
   root(N,R,Rp),
   Tp is T/N,
   cartesien([Rp,Tp],Y).

root(N,X,Y) :- Y is exp(log(X)/N).

polaire([X,Y],[R,Tp]) :-
   R is sqrt(X**2+Y**2),
   T is acos(abs(X)/R),
   cadran(X,Y,T,Tp).

cadran(X,Y,T,Tp) :- X >= 0, Y >= 0, Tp = T.
cadran(X,Y,T,Tp) :- X < 0, Y >= 0, Tp is pi-T.
cadran(X,Y,T,Tp) :- X < 0, Y < 0, Tp is T+pi.
cadran(X,Y,T,Tp) :- X >= 0, Y < 0, Tp is 2*pi-T.

cartesien([R,T],[X1,X2]) :- X1 is R*cos(T), X2 is R*sin(T).

/* Problemes de zero */

nul([X,Y])   :- nulreel(X), nulreel(Y), !.

nonnul(Z) :- nul(Z), !, fail.
nonnul(Z).

nulreel(0) :- !.  nulreel(0.0) :- !. nulreel(-0.0).
```